FDTD 法入門

平野拓一

FDTD (Finite Difference Time Domain method) 法とは、マクスウェルの方程式を差分化して電磁界をシミュレートする方法である。

1. マクスウェルの方程式

$$\nabla \times \mathbf{E} + \frac{\partial \mathbf{B}}{\partial t} = 0 \tag{1}$$

$$\nabla \times \mathbf{H} - \frac{\partial \mathbf{D}}{\partial t} = \mathbf{i} \tag{2}$$

媒質は等方・非分散とし、構成方程式 $\mathbf{B} = \mu \mathbf{H}, \mathbf{D} = \varepsilon \mathbf{E}, \mathbf{i} = \sigma \mathbf{E}$ を代入すると

$$\nabla \times \mathbf{E} + \mu \frac{\partial \mathbf{H}}{\partial t} = 0 \tag{3}$$

$$\nabla \times \mathbf{H} - \varepsilon \frac{\partial \mathbf{E}}{\partial t} = \sigma \mathbf{E} \tag{4}$$

2. 差分方程式

2.1 時間の離散化

$$\frac{\partial \mathbf{H}}{\partial t} = -\frac{1}{\mu} \nabla \times \mathbf{E}$$

$$\frac{\partial \mathbf{E}}{\partial t} = \frac{1}{\varepsilon} \nabla \times \mathbf{H} - \frac{\sigma}{\varepsilon} \mathbf{E}$$

時間を離散化する。

$$\frac{\mathbf{H}^{n+\frac{1}{2}} - \mathbf{H}^{n-\frac{1}{2}}}{\Delta t} = -\frac{1}{\mu} \nabla \times \mathbf{E}^n$$

$$\frac{\mathbf{E}^n - \mathbf{E}^{n-1}}{\Delta t} = -\frac{\sigma}{\varepsilon} \mathbf{E}^{n-\frac{1}{2}} + \frac{1}{\varepsilon} \nabla \times \mathbf{H}^{n-\frac{1}{2}}$$

ここで、$\mathbf{E}^{n-\frac{1}{2}}$ はないので $\dfrac{\mathbf{E}^n + \mathbf{E}^{n-1}}{2}$ で近似する(p.6)。

$$\frac{\mathbf{H}^{n+\frac{1}{2}} - \mathbf{H}^{n-\frac{1}{2}}}{\Delta t} = -\frac{1}{\mu} \nabla \times \mathbf{E}^n$$

$$\frac{\mathbf{E}^n-\mathbf{E}^{n-1}}{\Delta t}=-\frac{\sigma}{\varepsilon}\frac{\mathbf{E}^n+\mathbf{E}^{n-1}}{2}+\frac{1}{\varepsilon}\nabla\times\mathbf{H}^{n-\frac{1}{2}}$$

$\mathbf{E}^n,\mathbf{H}^{n+\frac{1}{2}}$について解くと、

$$\mathbf{E}^n-\mathbf{E}^{n-1}=-\frac{\sigma\Delta t}{\varepsilon}\frac{\mathbf{E}^n+\mathbf{E}^{n-1}}{2}+\frac{\Delta t}{\varepsilon}\nabla\times\mathbf{H}^{n-\frac{1}{2}}$$

$$\left(1+\frac{\sigma\Delta t}{2\varepsilon}\right)\mathbf{E}^n=\left(1-\frac{\sigma\Delta t}{2\varepsilon}\right)\mathbf{E}^{n-1}+\frac{\Delta t}{\varepsilon}\nabla\times\mathbf{H}^{n-\frac{1}{2}}$$

$$\mathbf{E}^n=\frac{1-\dfrac{\sigma\Delta t}{2\varepsilon}}{1+\dfrac{\sigma\Delta t}{2\varepsilon}}\mathbf{E}^{n-1}+\frac{\Delta t/\varepsilon}{1+\dfrac{\sigma\Delta t}{2\varepsilon}}\nabla\times\mathbf{H}^{n-\frac{1}{2}}$$

$$\mathbf{H}^{n+\frac{1}{2}}-\mathbf{H}^{n-\frac{1}{2}}=-\frac{\Delta t}{\mu}\nabla\times\mathbf{E}^n$$

$$\mathbf{H}^{n+\frac{1}{2}}=\mathbf{H}^{n-\frac{1}{2}}-\frac{\Delta t}{\mu}\nabla\times\mathbf{E}^n$$

$$\boxed{\begin{aligned}\mathbf{E}^n&=\frac{1-\dfrac{\sigma\Delta t}{2\varepsilon}}{1+\dfrac{\sigma\Delta t}{2\varepsilon}}\mathbf{E}^{n-1}+\frac{\Delta t/\varepsilon}{1+\dfrac{\sigma\Delta t}{2\varepsilon}}\nabla\times\mathbf{H}^{n-\frac{1}{2}}\\ \mathbf{H}^{n+\frac{1}{2}}&=\mathbf{H}^{n-\frac{1}{2}}-\frac{\Delta t}{\mu}\nabla\times\mathbf{E}^n\end{aligned}}$$

2.2 位置の離散化

図 1 のように空間を離散化して電磁界を配置すると、電界と磁界がうまくずれているので互いにローテーションを取ることができる。また、図 2 は図 1 の単位セルを表していて、このように電界、磁界を配置したセルを **Yee セル**と言う。同じ場所に電界と磁界を配置していないので電界、磁界を時間更新していくときに電界と磁界を交互に更新し、その結果過去の値を保持する必要もなくなり、メモリを節約できる。これを **Leap-flog (蛙飛び)** アルゴリズムと言う。

例えば、ある場所の電界はその場所の 1 ステップ前の電界と、その場所の磁界のローテーションから計算される。同様に、ある場所の磁界はその場所の 1 ステップ前の磁界と、その場所の電界のローテーションから計算される。



図 1 セル分割された解析領域



図 2 単位セル

$\nabla \times \mathbf{A}$ の計算

$$\nabla \times \mathbf{A} = \begin{vmatrix} \hat{x} & \hat{y} & \hat{z} \\ \partial / \partial x & \partial / \partial y & \partial / \partial z \\ A_x & A_y & A_z \end{vmatrix}$$
$$= \hat{x}\left(\frac{\partial A_z}{\partial y} - \frac{\partial A_y}{\partial z}\right) - \hat{y}\left(\frac{\partial A_z}{\partial x} - \frac{\partial A_x}{\partial z}\right) + \hat{z}\left(\frac{\partial A_y}{\partial x} - \frac{\partial A_x}{\partial y}\right)$$

次の表現に現れる $(i+\frac{1}{2}, j, k), (i, j+\frac{1}{2}, k), (i, j, k+\frac{1}{2})$ などはプログラム中では全て (i, j, k) として配列に格納すればよい。

電界（p.27）

$$E_x^n(i+\frac{1}{2}, j, k) = C_{ex}(i+\frac{1}{2}, j, k) E_x^{n-1}(i+\frac{1}{2}, j, k)$$
$$+ C_{exry}(i+\frac{1}{2}, j, k)\left\{H_z^{n-\frac{1}{2}}(i+\frac{1}{2}, j+\frac{1}{2}, k) - H_z^{n-\frac{1}{2}}(i+\frac{1}{2}, j-\frac{1}{2}, k)\right\}$$
$$- C_{exrz}(i+\frac{1}{2}, j, k)\left\{H_y^{n-\frac{1}{2}}(i+\frac{1}{2}, j, k+\frac{1}{2}) - H_y^{n-\frac{1}{2}}(i+\frac{1}{2}, j, k-\frac{1}{2})\right\}$$

$$E_y^n(i, j+\frac{1}{2}, k) = C_{ey}(i, j+\frac{1}{2}, k) E_y^{n-1}(i, j+\frac{1}{2}, k)$$
$$- C_{eyrx}(i, j+\frac{1}{2}, k)\left\{H_z^{n-\frac{1}{2}}(i+\frac{1}{2}, j+\frac{1}{2}, k) - H_z^{n-\frac{1}{2}}(i-\frac{1}{2}, j+\frac{1}{2}, k)\right\}$$
$$+ C_{eyrz}(i, j+\frac{1}{2}, k)\left\{H_x^{n-\frac{1}{2}}(i, j+\frac{1}{2}, k+\frac{1}{2}) - H_x^{n-\frac{1}{2}}(i, j+\frac{1}{2}, k-\frac{1}{2})\right\}$$

$$E_z^n(i, j, k+\frac{1}{2}) = C_{ez}(i, j, k+\frac{1}{2}) E_y^{n-1}(i, j, k+\frac{1}{2})$$
$$- C_{ezrx}(i, j, k+\frac{1}{2})\left\{H_y^{n-\frac{1}{2}}(i+\frac{1}{2}, j, k+\frac{1}{2}) - H_y^{n-\frac{1}{2}}(i-\frac{1}{2}, j, k+\frac{1}{2})\right\}$$
$$+ C_{ezry}(i, j, k+\frac{1}{2})\left\{H_x^{n-\frac{1}{2}}(i, j+\frac{1}{2}, k+\frac{1}{2}) - H_x^{n-\frac{1}{2}}(i, j-\frac{1}{2}, k+\frac{1}{2})\right\}$$

ただし、

$$C_{ex}(i+\frac{1}{2},j,k)=\frac{1-\dfrac{\sigma(i+\frac{1}{2},j,k)\Delta t}{2\varepsilon(i+\frac{1}{2},j,k)}}{1+\dfrac{\sigma(i+\frac{1}{2},j,k)\Delta t}{2\varepsilon(i+\frac{1}{2},j,k)}}$$

$$C_{exrz}(i+\frac{1}{2},j,k)=\frac{\Delta t/\varepsilon(i+\frac{1}{2},j,k)}{1+\dfrac{\sigma(i+\frac{1}{2},j,k)\Delta t}{2\varepsilon(i+\frac{1}{2},j,k)}}\frac{1}{\Delta z}$$

$$C_{exry}(i+\frac{1}{2},j,k)=\frac{\Delta t/\varepsilon(i+\frac{1}{2},j,k)}{1+\dfrac{\sigma(i+\frac{1}{2},j,k)\Delta t}{2\varepsilon(i+\frac{1}{2},j,k)}}\frac{1}{\Delta y}$$

$$C_{ey}(i,j+\frac{1}{2},k)=\frac{1-\dfrac{\sigma(i,j+\frac{1}{2},k)\Delta t}{2\varepsilon(i,j+\frac{1}{2},k)}}{1+\dfrac{\sigma(i,j+\frac{1}{2},k)\Delta t}{2\varepsilon(i,j+\frac{1}{2},k)}}$$

$$C_{eyrz}(i,j+\frac{1}{2},k)=\frac{\Delta t/\varepsilon(i,j+\frac{1}{2},k)}{1+\dfrac{\sigma(i,j+\frac{1}{2},k)\Delta t}{2\varepsilon(i,j+\frac{1}{2},k)}}\frac{1}{\Delta z}$$

$$C_{eyrx}(i,j+\frac{1}{2},k)=\frac{\Delta t/\varepsilon(i,j+\frac{1}{2},k)}{1+\dfrac{\sigma(i,j+\frac{1}{2},k)\Delta t}{2\varepsilon(i,j+\frac{1}{2},k)}}\frac{1}{\Delta x}$$

$$C_{ez}(i,j,k+\frac{1}{2})=\frac{1-\dfrac{\sigma(i,j,k+\frac{1}{2})\Delta t}{2\varepsilon(i,j,k+\frac{1}{2})}}{1+\dfrac{\sigma(i,j,k+\frac{1}{2})\Delta t}{2\varepsilon(i,j,k+\frac{1}{2})}}$$

$$C_{ezrx}(i,j,k+\frac{1}{2})=\frac{\Delta t/\varepsilon(i,j,k+\frac{1}{2})}{1+\dfrac{\sigma(i,j,k+\frac{1}{2})\Delta t}{2\varepsilon(i,j,k+\frac{1}{2})}}\frac{1}{\Delta x}$$

$$C_{ezry}(i,j,k+\frac{1}{2})=\frac{\Delta t/\varepsilon(i,j,k+\frac{1}{2})}{1+\dfrac{\sigma(i,j,k+\frac{1}{2})\Delta t}{2\varepsilon(i,j,k+\frac{1}{2})}}\frac{1}{\Delta y}$$

電界の更新プログラムは次のようになる。i,j,k のインデックスの始まりが 1 または 2 となっているのは図 1 において色付けされた壁は後で述べる吸収境界壁として使うから、その上にのっている電磁界を更新しないからである。

```
! Ex
do k=2,nz-1
    do j=2,ny-1
        do i=1,nx-1
            id=media_id(i,j,k)

            if(id.eq.1) then
                ! 1: 自由空間
                ex(i,j,k)=cex0*ex(i,j,k) &
                            +cexry0*(hz(i,j,k)-hz(i,j-1,k)) &
                            -cexrz0*(hy(i,j,k)-hy(i,j,k-1))
            else if(id.eq.2) then
                ! 2: 完全導体
                ex(i,j,k)=0.0d0
            else
                ! 3 以上: 任意媒質
                ex(i,j,k)=cex(id)*ex(i,j,k) &
                            +cexry(id)*(hz(i,j,k)-hz(i,j-1,k)) &
                            -cexrz(id)*(hy(i,j,k)-hy(i,j,k-1))
            end if
```

```
    end do
end do
        end do

! Ey
do k=2,nz-1
    do j=1,ny-1
        do i=2,nx-1
            id=media_id(i,j,k)

            if(id.eq.1) then
                ! 1: 自由空間
                ey(i,j,k)=cey0*ey(i,j,k) &
                            +ceyrz0*(hx(i,j,k)-hx(i,j,k-1)) &
                            -ceyrx0*(hz(i,j,k)-hz(i-1,j,k))
            else if(id.eq.2) then
                ! 2: 完全導体
                ey(i,j,k)=0.0d0
            else
                ! 3 以上: 任意媒質
                ey(i,j,k)=cey(id)*ey(i,j,k) &
                            +ceyrz(id)*(hx(i,j,k)-hx(i,j,k-1)) &
                            -ceyrx(id)*(hz(i,j,k)-hz(i-1,j,k))
            end if
        end do
    end do
end do

! Ez
do k=1,nz-1
    do j=2,ny-1
        do i=2,nx-1
            id=media_id(i,j,k)

            if(id.eq.1) then
                ! 1: 自由空間
                ez(i,j,k)=cez0*ez(i,j,k) &
                            +cezrx0*(hy(i,j,k)-hy(i-1,j,k)) &
                            -cezry0*(hx(i,j,k)-hx(i,j-1,k))
            else if(id.eq.2) then
                ! 2: 完全導体
                ez(i,j,k)=0.0d0
            else
                ! 3 以上: 任意媒質
                ez(i,j,k)=cez(id)*ez(i,j,k) &
                            +cezrx(id)*(hy(i,j,k)-hy(i-1,j,k)) &
                            -cezry(id)*(hx(i,j,k)-hx(i,j-1,k))
            end if
        end do
    end do
end do
```

磁界（p.29）

$$H_x^{n+\frac{1}{2}}(i,j+\frac{1}{2},k+\frac{1}{2})=H_x^{n-\frac{1}{2}}(i,j+\frac{1}{2},k+\frac{1}{2})$$
$$-C_{hxry}(i,j+\frac{1}{2},k+\frac{1}{2})\left\{E_z^n(i,j+1,k+\frac{1}{2})-E_z^n(i,j,k+\frac{1}{2})\right\}$$
$$+C_{hxrz}(i,j+\frac{1}{2},k+\frac{1}{2})\left\{E_y^n(i,j+\frac{1}{2},k+1)-E_y^n(i,j+\frac{1}{2},k)\right\}$$

$$H_y^{n+\frac{1}{2}}(i+\frac{1}{2},j,k+\frac{1}{2})=H_y^{n-\frac{1}{2}}(i+\frac{1}{2},j,k+\frac{1}{2})$$
$$-C_{hyrz}(i+\frac{1}{2},j,k+\frac{1}{2})\left\{E_x^n(i+\frac{1}{2},j,k+1)-E_x^n(i+\frac{1}{2},j,k)\right\}$$
$$+C_{hyrx}(i+\frac{1}{2},j,k+\frac{1}{2})\left\{E_z^n(i+1,j,k+\frac{1}{2})-E_z^n(i,j,k+\frac{1}{2})\right\}$$

$$H_z^{n+\frac{1}{2}}(i+\frac{1}{2},j+\frac{1}{2},k)=H_z^{n-\frac{1}{2}}(i+\frac{1}{2},j+\frac{1}{2},k)$$
$$-C_{hyrx}(i+\frac{1}{2},j+\frac{1}{2},k)\left\{E_y^n(i+1,j+\frac{1}{2},k)-E_y^n(i,j+\frac{1}{2},k)\right\}$$
$$+C_{hyry}(i+\frac{1}{2},j+\frac{1}{2},k)\left\{E_x^n(i+\frac{1}{2},j+1,k)-E_x^n(i+\frac{1}{2},j,k)\right\}$$

ただし、

$$C_{hxry}(i,j+\frac{1}{2},k+\frac{1}{2})=\frac{\Delta t}{\mu(i,j+\frac{1}{2},k+\frac{1}{2})}\frac{1}{\Delta y}$$
$$C_{hxrz}(i,j+\frac{1}{2},k+\frac{1}{2})=\frac{\Delta t}{\mu(i,j+\frac{1}{2},k+\frac{1}{2})}\frac{1}{\Delta z}$$

$$C_{hyrz}(i+\frac{1}{2},j,k+\frac{1}{2})=\frac{\Delta t}{\mu(i+\frac{1}{2},j,k+\frac{1}{2})}\frac{1}{\Delta z}$$
$$C_{hyrx}(i+\frac{1}{2},j,k+\frac{1}{2})=\frac{\Delta t}{\mu(i+\frac{1}{2},j,k+\frac{1}{2})}\frac{1}{\Delta x}$$

$$C_{hzrx}(i+\frac{1}{2},j+\frac{1}{2},k)=\frac{\Delta t}{\mu(i+\frac{1}{2},j+\frac{1}{2},k)}\frac{1}{\Delta x}$$

$$C_{hzry}(i+\frac{1}{2},j+\frac{1}{2},k)=\frac{\Delta t}{\mu(i+\frac{1}{2},j+\frac{1}{2},k)}\frac{1}{\Delta y}$$

磁界に対するプログラムコードも電界の場合と同様にして次のようになる。

```
! Hx
do k=1,nz-1
    do j=1,ny-1
        do i=2,nx-1
            id=media_id(i,j,k)

            if(id.eq.1) then
                ! 1: 自由空間
                hx(i,j,k)=hx(i,j,k) &
                            -chxry0*(ez(i,j+1,k)-ez(i,j,k)) &
                            +chxrz0*(ey(i,j,k+1)-ey(i,j,k))
            else if(id.eq.2) then
                ! 2: 完全導体
                hx(i,j,k)=0.0d0
            else
                ! 3 以上: 任意媒質
                hx(i,j,k)=hx(i,j,k) &
                            -chxry(id)*(ez(i,j+1,k)-ez(i,j,k)) &
                            +chxrz(id)*(ey(i,j,k+1)-ey(i,j,k))
            end if
        end do
    end do
end do

! Hy
do k=1,nz-1
    do j=2,ny-1
        do i=1,nx-1
            id=media_id(i,j,k)

            if(id.eq.1) then
                ! 1: 自由空間
                hy(i,j,k)=hy(i,j,k) &
                            -chyrz0*(ex(i,j,k+1)-ex(i,j,k)) &
                            +chyrx0*(ez(i+1,j,k)-ez(i,j,k))
            else if(id.eq.2) then
                ! 2: 完全導体
                hy(i,j,k)=0.0d0
            else
```

```
               hy(i,j,k)=hy(i,j,k) &
                           -chyrz(id)*(ex(i,j,k+1)-ex(i,j,k)) &
               ! 3 以上: 任意媒質
                           +chyrx(id)*(ez(i+1,j,k)-ez(i,j,k))
            end if
         end do
      end do
   end do

   ! Hz
   do k=2,nz-1
      do j=1,ny-1
         do i=1,nx-1
            id=media_id(i,j,k)

            if(id.eq.1) then
               ! 1: 自由空間
               hz(i,j,k)=hz(i,j,k) &
                           -chzrx0*(ey(i+1,j,k)-ey(i,j,k)) &
                           +chzry0*(ex(i,j+1,k)-ex(i,j,k))
            else if(id.eq.2) then
               ! 2: 完全導体
               hz(i,j,k)=0.0d0
            else
               ! 3 以上: 任意媒質
               hz(i,j,k)=hz(i,j,k) &
                           -chzrx(id)*(ey(i+1,j,k)-ey(i,j,k)) &
                           +chzry(id)*(ex(i,j+1,k)-ex(i,j,k))
            end if
         end do
      end do
   end do
```

3. モデリング

図 1の解析領域に対して、媒質定数(ε,μ,σ)を設定してモデリングを行う。

媒質の種類があまり多くない場合、各セルに媒質番号（整数）のインデックスを与えて処理した方が、各セルに媒質乗数（実数）を与えるよりもメモリを節約できる。よって、各セルに媒質のインデックスを与えて電磁界計算の処理を行う。

電磁界の更新には媒質定数も使われるが、それがそのまま更新式にある訳ではなく、更新係数によって間接的に行われるので、プログラムの最初の部分で各媒質に応じた電磁界の更新係数をあらかじめ計算しておく。

```
! ******** 媒質定数設定 ********
nmedia=3
! 1: 真空
```

```
mu(1)=mu0
sig(1)=0.0d0
eps(1)=epsilon0
! 2: PEC,PMC (これは別のルーチンで処理する)
! 3: 誘電体（テフロン）
eps(1)=2.17d0*epsilon0
mu(1)=mu0
sig(1)=0.0d0

! ******** フィールド更新係数 ********
! 1: 真空
cex0=1.0d0
cey0=1.0d0
cez0=1.0d0

cexry0=(dt/epsilon0)/dy
cexrz0=(dt/epsilon0)/dz
ceyrz0=(dt/epsilon0)/dz
ceyrx0=(dt/epsilon0)/dx
cezrx0=(dt/epsilon0)/dx
cezry0=(dt/epsilon0)/dy

chxry0=(dt/mu0)/dy
chxrz0=(dt/mu0)/dz
chyrz0=(dt/mu0)/dz
chyrx0=(dt/mu0)/dx
chzrx0=(dt/mu0)/dx
chzry0=(dt/mu0)/dy

! 3 以上: 一般の媒質
do id=3,nmedia
   cex(id)=(1.0d0-((sig(id)*dt)/(2.0d0*eps(id)))) &
           /(1.0d0+((sig(id)*dt)/(2.0d0*eps(id))))
   cey(id)=(1.0d0-((sig(id)*dt)/(2.0d0*eps(id)))) &
           /(1.0d0+((sig(id)*dt)/(2.0d0*eps(id))))
   cez(id)=(1.0d0-((sig(id)*dt)/(2.0d0*eps(id)))) &
           /(1.0d0+((sig(id)*dt)/(2.0d0*eps(id))))

   cexry(id)=(dt/eps(id))/(1.0d0+((sig(id)*dt)/(2.0d0*eps(id))))/dy
   cexrz(id)=(dt/eps(id))/(1.0d0+((sig(id)*dt)/(2.0d0*eps(id))))/dz
   ceyrz(id)=(dt/eps(id))/(1.0d0+((sig(id)*dt)/(2.0d0*eps(id))))/dz
   ceyrx(id)=(dt/eps(id))/(1.0d0+((sig(id)*dt)/(2.0d0*eps(id))))/dx
   cezrx(id)=(dt/eps(id))/(1.0d0+((sig(id)*dt)/(2.0d0*eps(id))))/dx
   cezry(id)=(dt/eps(id))/(1.0d0+((sig(id)*dt)/(2.0d0*eps(id))))/dy

   chxry(id)=(dt/mu(id))/dy
   chxrz(id)=(dt/mu(id))/dz
   chyrz(id)=(dt/mu(id))/dz
   chyrx(id)=(dt/mu(id))/dx
   chzrx(id)=(dt/mu(id))/dx
   chzry(id)=(dt/mu(id))/dy
end do
```

4. 波源

4.1 電流源

$$\nabla\times\mathbf{H}-\frac{\partial\mathbf{D}}{\partial t}=\mathbf{i}$$

$\mathbf{i}=\sigma\mathbf{E}+\mathbf{J}$ とすると

$$\frac{\partial\mathbf{D}}{\partial t}=\nabla\times\mathbf{H}-\mathbf{i}=\nabla\times\mathbf{H}-\sigma\mathbf{E}-\mathbf{J}$$

$$\frac{\partial\mathbf{E}}{\partial t}=-\frac{\sigma}{\varepsilon}\mathbf{E}+\frac{1}{\varepsilon}\nabla\times\mathbf{H}-\frac{1}{\varepsilon}\mathbf{J}$$

2.1 のように離散化すると

$$\mathbf{E}^{n}=\frac{1-\frac{\sigma\Delta t}{2\varepsilon}}{1+\frac{\sigma\Delta t}{2\varepsilon}}\mathbf{E}^{n-1}+\frac{\Delta t/\varepsilon}{1+\frac{\sigma\Delta t}{2\varepsilon}}\nabla\times\mathbf{H}^{n-\frac{1}{2}}-\frac{\Delta t/\varepsilon}{1+\frac{\sigma\Delta t}{2\varepsilon}}\mathbf{J}^{n-\frac{1}{2}}$$

4.2 磁流源

磁流を考慮した Maxwell の方程式は

$$\nabla\times\mathbf{E}+\frac{\partial\mathbf{B}}{\partial t}=-\mathbf{m}$$

$$\nabla\times\mathbf{H}-\frac{\partial\mathbf{D}}{\partial t}=\mathbf{i}$$

となり、3.1 と同様に計算を行う。$\mathbf{E}\to-\mathbf{H}$, $\mathbf{H}\to\mathbf{E}$, $\mathbf{i}\to\mathbf{m}$, $\varepsilon\to\mu$, $\sigma\to 0$ と書き換えればよいだけなので

$$\mathbf{H}^{n}=\mathbf{H}^{n-1}-\frac{\Delta t}{\mu}\nabla\times\mathbf{E}^{n-\frac{1}{2}}+\frac{\Delta t}{\mu}\mathbf{m}^{n-\frac{1}{2}}$$

4.3 平面波

平面波が 1 方向に進むようにするために電界と磁界を界インピーダンスの比にして、ある面に対して与える。

5. 吸収境界条件

[参考文献]

[1] 宇野亨: 「FDTD 法による電磁界およびアンテナ解析」, コロナ社, 1998